
Abstract

Logic Programs with Ordered Disjunction (LPOD) is an extension of standard answer set programs to
handle preference using the construct of ordered disjunction, and CR-Prolog2 is an extension of standard
answer set programs with consistency restoring rules and LPOD-like ordered disjunction. We present re-
ductions of each of these languages into the standard ASP language, which gives us an alternative way to
understand the extensions in terms of the standard ASP language.

1 Introduction

In answer set programming, each answer set encodes a solution to the problem that is being
modeled. There is often a need to express that one solution is preferable to another, so several ex-
tensions of answer set programs were made to express a qualitative preference over answer sets.
In Logic Programs with Ordered Disjunction (LPOD) (Brewka 2002), this is done by introduc-
ing the construct of ordered disjunction in the head of a rule: A×B ← Body intuitively means,
when Body is true, if possible then A, but if A is not possible, then at least B. Proposition 2 from
(Brewka 2002) states that there is no reduction of LPOD to disjunctive logic programs (Gel-
fond and Lifschitz 1991) based on the fact that the answer sets of disjunctive logic programs are
subset-minimal whereas LPOD answer sets are not necessarily so. However, this justification is
limited to translations that preserve the underlying signature, and it remained an open question
if it is possible to turn LPOD into the language of standard ASP such as ASP-Core 2 (Calimeri
et al. 2012) by using auxiliary atoms. In this paper, we provide a positive answer to this question.

We present a reduction of LPOD to standard answer set programs by compiling away ordered
disjunctions. The translation gives us an alternative way to understand the semantics of LPOD in
terms of the standard ASP language, and more generally, a method to express preference relations
among answer sets. Instead of iterating the generator and the tester programs as in (Brewka et al.
2002), our reduction is one-pass: the preferred answer sets can be computed by calling an answer
set solver one time.

It turns out that the translation idea is not restricted to LPOD but also applies to CR-Prolog2

(Balduccini and Mellarkod 2004), which not only has a construct similar to ordered disjunction
in LPOD but also inherits the construct of consistency-restoring rules—rules that can be added to
make inconsistent programs to be consistent—from CR-Prolog (Balduccini and Gelfond 2003).
With some modifications to the LPOD translation, we show that CR-Prolog2 programs can also



2 Lee & Yang

be turned into standard answer set programs by compiling away both ordered disjunctions and
consistency-restoring rules.

The paper is organized as follows. Section 2 reviews LPOD and presents a translation that
turns LPOD into standard answer set programs. Section 3 reviews CR-Prolog2 and presents a
translation that turns CR-Prolog2 into standard answer set programs. The complete proofs are in
the supplementary material at the TPLP archives (Lee and Yang 2018).

2 LPOD to ASP with Weak Constraints

2.1 Review: LPOD

We review the definition of LPOD by Brewka (2002). As in that paper, for simplicity, we assume
the underlying signature is propositional.

Syntax: A (propositional) LPOD Π is Πreg ∪Πod, where its regular part Πreg consists of usual
ASP rules Head ← Body, and its ordered disjunction part Πod consists of LPOD rules of the
form

C1 × · · · × Cn ← Body (1)

in which Ci are atoms, n is at least 2, and Body is a conjunction of atoms possibly preceded by
not.1 Rule (1) intuitively says “when Body is true, if possible then C1; if C1 is not possible then
C2; . . . ; if all of C1, . . . , Cn−1 are not possible then Cn.”

Semantics: For an LPOD rule (1), its i-th option (i = 1, . . . , n) is defined as

Ci ← Body, not C1, . . . , not Ci−1.

A split program of an LPOD Π is obtained from Π by replacing each rule in Πod by one of its
options. A set S of atoms is a candidate answer set of Π if it is an answer set of a split program
of Π.

Example 1
(From (Brewka 2002)) The following LPOD Π1,

a× b ← not c
b× c ← not d,

has four split programs:

a← not c a← not c
b← not d c← not d, not b

b← not c, not a b← not c, not a
b← not d c← not d, not b.

(2)

Each of them has the following answer sets respectively, which are the candidate answer sets
of Π1.

{a, b} {c}

{b} {b}, {c}.

1 In (Brewka 2002), a usual ASP rule is viewed as a special case of a rule with ordered disjunction when n = 1 but in
this paper, we distinguish them. This simplifies the presentation of the translation and also allows us to consider LPOD
that are more general than the original definition by allowing modern ASP constructs such as aggregates.

A candidate answer set S of Π is said to satisfy rule (1)

• to degree 1 if S does not satisfy Body, and
• to degree j (1 ≤ j ≤ n) if S satisfies Body and j = min{k | Ck ∈ S}.

When Πod contains m LPOD rules, the satisfaction degree list of a candidate answer set S
of Π is (d1, . . . , dm) where di is the degree to which S satisfies rule i in Πod. For a candidate
answer set S, let Si(Π) denote the set of rules in Πod satisfied by S to degree i. For candidate
answer sets S1 and S2 of Π, Brewka (2005) introduces the following four preference criteria.

1. Cardinality-Preferred: S1 is cardinality-preferred to S2 (S1 >
c S2) if there is a positive

integer i such that |Si
1(Π)| > |Si

2(Π)|, and |Sj
1(Π)| = |Sj

2(Π)| for all j < i.
2. Inclusion-Preferred: S1 is inclusion-preferred to S2 (S1 >i S2) if there is a positive

integer i such that Si
2(Π) ⊂ Si

1(Π), and Sj
1(Π) = Sj

2(Π) for all j < i.
3. Pareto-Preferred: S1 is Pareto-preferred to S2 (S1 >p S2) if there is a rule that is

satisfied to a lower degree in S1 than in S2, and there is no rule that is satisfied to a lower
degree in S2 than in S1.

4. Penalty-Sum-Preferred: S1 is penalty-sum-preferred to S2 (S1 >
ps S2) if the sum of

the satisfaction degrees of all rules is smaller in S1 than in S2.

A candidate answer set S of Π is a k-preferred (k ∈ {c, i, p, ps}) answer set if there is no
candidate answer set S′ of Π such that S′ >k S.

Example 1 (Continued)
Recall that Π1 has three candidate answer sets: {a, b}, {b}, and {c}. Their satisfaction degree lists
are (1,1), (2,1), and (1,2), respectively. One can check that {a, b} is the only preferred answer set
according to any of the four preference criteria.

Example 2
To illustrate the difference among the four preference criteria, consider the following LPOD Π2

about picking a hotel near the Grand Canyon. hotel(1) is a 2 star hotel but is close to the Grand
Canyon, hotel(2) is a 3 star hotel and the distance is medium, and hotel(3) is a 4 star hotel but
is too far.

close×med× far × tooFar
star4× star3× star2
1{hotel(X) : X = 1..3}1
⊥ ← hotel(1), not close
⊥ ← hotel(1), not star2

⊥ ← hotel(2), not med
⊥ ← hotel(2), not star3
⊥ ← hotel(3), not tooFar
⊥ ← hotel(3), not star4

Π2 has 4 × 3 split programs but only the following three programs are consistent (The regular
part of Π2 is not listed).

close med← not close
star2← not star4, not star3 star3← not star4

tooFar ← not close, not med, not far
star4

The candidate answer sets of Π2 and their satisfaction degree lists are

S1 = {hotel(1), close, star2, . . . }, (1, 3) S2 = {hotel(2),med, star3, . . . }, (2, 2)

S3 = {hotel(3), tooFar, star4, . . . }, (4, 1)

By definition, the cardinality-preferred answer set is S1, the inclusion-preferred answer sets are
S1 and S3, the Pareto-preferred answer sets are S1, S2 and S3, while the penalty-sum-preferred
answer sets are S1 and S2.

2.2 An Alternative Way to Generate Candidate Answer Sets: Assumption Programs

Before we describe the translation of LPOD into standard answer set programs, we consider
an alternative way to generate candidate answer sets together with their “assumption degrees,”
which serves as a basis of our translation.

Let Π be an LPOD with m LPOD rules. For an LPOD rule i (i ∈ {1, . . . ,m})

C1
i × · · · × C

ni
i ← Bodyi , (3)

its x-th assumption (x ∈ {0, . . . , ni}), denoted by Oi(x), is defined as the set of ASP rules

bodyi ← Bodyi (4)

⊥ ← x = 0, bodyi (5)

⊥ ← x > 0, not bodyi (6)

Cj
i ← bodyi, x = j (for 1 ≤ j ≤ ni) (7)

⊥ ← bodyi, x 6= j, not C1
i , . . . , not Cj−1

i , Cj
i (for 1 ≤ j ≤ ni) (8)

where bodyi is a new, distinct atom for each LPOD rule i. Rules (4)—(6) ensure that the body of
(3) is false iff x = 0. Rule (7) represents that Cx

i is true under the x-th assumption, and rule (8)
ensures that all atoms C1

i , . . . , C
x−1
i are false. The last two rules together tells us that the first

atom in C1
i , . . . , C

ni
i that is true is Cx

i . The reason we call rules (4)—(8) the x-th assumption is
because they encode a certain assumption imposed on rule (3) in deriving each candidate answer
set: x = 0 assumes Bodyi is false, whereas x > 0 assumes Bodyi is true and the x-th atom in the
head is to be derived.

An assumption program of an LPOD Π is obtained from Π by replacing each rule in Πod

by one of its assumptions. If each LPOD rule i is replaced by its xi-th assumption, we call
(x1, . . . , xm) the assumption degree list of the assumption program.

The following proposition asserts that the candidate answer sets can be obtained from assump-
tion programs instead of split programs.

Proposition 1
For any LPOD Π of σ and any set S of atoms of σ, S is a candidate answer set of Π iff S∪{bodyi |
S satisfies the body of rule i in Πod} is an answer set of some assumption program of Π.

Example 1 (Continued) The assumptions for rule a × b ← not c, denoted by O1(X1), and the
assumptions for rule b× c← not d, denoted by O2(X2) are as follows, where X1 and X2 range
over {0, 1, 2}.

O1(X1) : body1 ← not c O2(X2) : body2 ← not d
⊥ ← X1 = 0, body1 ⊥ ← X2 = 0, body2
⊥ ← X1 > 0, not body1 ⊥ ← X2 > 0, not body2
a ← body1, X1=1 b ← body2, X2=1
b ← body1, X1=2 c ← body2, X2=2
⊥ ← body1, X1 6= 1, a ⊥ ← body2, X2 6= 1, b
⊥ ← body1, X1 6= 2, not a, b ⊥ ← body2, X2 6= 2, not b, c

Π1 has 9 assumption programs,

O1(0) ∪O2(0) O1(0) ∪O2(1) O1(0) ∪O2(2) , {c}

O1(1) ∪O2(0) O1(1) ∪O2(1) , {a, b} O1(1) ∪O2(2)

O1(2) ∪O2(0) O1(2) ∪O2(1) , {b} O1(2) ∪O2(2),

among which the three assumption programs in the boxes are consistent. Their answer sets are
shown together.

An advantage of considering assumption programs over split programs is that the satisfaction
degrees—a basis of comparing the candidate answer sets—can be obtained from the assumption
degrees with a minor modification (Section 2.3.1). This is in part because each candidate answer
set is obtained from only one assumption program whereas the same candidate answer set can be
obtained from multiple split programs (e.g., {b} in Example 1).

2.3 Turning LPOD into Standard Answer Set Programs

We define a translation lpod2asp(Π) that turns an LPOD Π into a standard answer set program.
Let Π be an LPOD of signature σ where Πod contains m propositional rules with ordered

disjunction:

1 : C1
1 × · · · × C

n1
1 ← Body1

. . . (9)

m : C1
m × · · · × Cnm

m ← Bodym

where 1, . . . ,m are rule indices, and ni ≥ 2 for 1 ≤ i ≤ m.
The first-order signature σ′ of lpod2asp(Π) contains m-ary predicate constant a/m for each

propositional constant a of σ. Besides, σ′ contains the following predicate constants not in σ:
ap/m (“assumption program”), degree/(m+1), bodyi/m (i ∈ {1, . . . ,m}), prf/2 (“preferred”),
and pAS/m (“preferred answer set”). Furthermore, σ′ contains the following predicate constants
according to each preference criterion:

• for cardinality-preferred: card/3, equ2degree/3, prf2degree/3

• for inclusion-preferred: even/1, equ2degree/3, prf2degree/3

• for Pareto-preferred: equ/2
• for penalty-sum-preferred: sum/2.

2.3.1 Generate Candidate Answer Sets

The first part of the translation lpod2asp(Π) is to generate all candidate answer sets of Π based
on the notion of assumption programs. We use the assumption degree list as a “name space” for
each candidate answer set, so that we can compare them in a single answer set program.

1. We use atom ap(x1, . . . , xm) to denote the assumption program whose assumption degree list
is (x1, . . . , xm). We consider all consistent assumption programs by generating a maximal set of
ap(·) atoms: ap(x1, . . . , xm) is included in an optimal answer set 2 iff the assumption program

2 For programs containing weak constraints, an optimal answer set is defined by the penalty that comes from the weak
constraints that are violated. (Calimeri et al. 2012)

denoted by ap(x1, . . . , xm) is consistent.

{ap(X1, . . . , Xm) : X1 = 0..n1, . . . , Xm = 0..nm}. (10)

:∼ ap(X1, . . . , Xm). [−1, X1, . . . , Xm] (11)

Rule (10) generates an arbitrary subset of ap(·) atoms, each of which records an assumption
degree list. Rule (11) is a weak constraint that maximizes the number of ap(·) atoms by adding
the penalty −1 for each true instance of ap(X1, . . . , Xm). Together with the rules below, these
rules ensure that we consider all assumption programs that are consistent and that no candidate
answer sets are missed in computing preference relationship in the second part of the translation.

2. We extend each atom to include the assumption degrees X1, . . . , Xm, and append atom
ap(X1, . . . , Xm) in the bodies of rules.

• For each rule Head ← Body in Πreg , lpod2asp(Π) contains

Head(X1, . . . , Xm)← ap(X1, . . . , Xm),Body(X1, . . . , Xm) (12)

where Head(X1, . . . , Xm) and Body(X1, . . . , Xm) are obtained from Head and Body by
replacing each atom A in them with A(X1, . . . , Xm). Each schematic variable Xi ranges
over {0, . . . , ni}.

• For each rule

C1
i × · · · × C

ni
i ← Bodyi

in Πod, where n ≥ 2, lpod2asp(Π) contains

bodyi(X1, . . . , Xm)← ap(X1, . . . , Xm),Bodyi(X1, . . . , Xm) (13)

⊥ ← ap(X1, . . . , Xm), Xi = 0, bodyi(X1, . . . , Xm) (14)

⊥ ← ap(X1, . . . , Xm), Xi > 0, not bodyi(X1, . . . , Xm). (15)

And for 1 ≤ j ≤ ni, lpod2asp(Π) contains

Cj
i (X1, . . . , Xm)← bodyi(X1, . . . , Xm), Xi = j. (16)

⊥ ← bodyi(X1, . . . , Xm), Xi 6= j,

not C1
i (X1, . . . , Xm), . . . , not Cj−1

i (X1, . . . , Xm), Cj
i (X1, . . . , Xm).

(17)

3. The satisfaction degree list can be obtained from the assumption degree list encoded in
ap(x1, . . . , xm) by changing xi to 1 if it was 0. For this, lpod2asp(Π) contains

1{degree(ap(X1, . . . , Xm), D1, . . . , Dm) : D1 = 1..n1, . . . , Dm = 1..nm}1
← ap(X1, . . . , Xm). (18)

and for 1 ≤ i ≤ m, lpod2asp(Π) contains

⊥ ← degree(ap(X1, . . . , Xm), D1, . . . , Dm), Xi = 0, Di 6= 1. (19)

⊥ ← degree(ap(X1, . . . , Xm), D1, . . . , Dm), Xi > 0, Di 6= Xi. (20)

Since all answer sets of the same assumption program are associated with the same satisfaction
degree list, we say an assumption program satisfies LPOD rule i to degree d if its answer sets
satisfy the rule to degree d. Rule (18) reads “for any assumption program ap(x1, . . . , xm), it has
exactly one assignment of satisfaction degrees D1, . . . , Dm.” Rules (19) and (20) say that the

assumption program ap(x1, . . . , xm) satisfies LPOD rule i to degree 1 if xi = 0 (in which case
Bodyi is false) and to degree xi if xi > 0 (in which case Bodyi is true).

Let us denote the set of rules (10)—(20) by lpod2asp(Π)base. Observe that the atoms a(v) in
the original signature σ are in the form of a(v, x1, . . . , xm) in the answer sets of lpod2asp(Π)base.
We define a way to retrieve the candidate answer set of Π by removing x1, . . . , xm as follows.
Let S be an optimal answer set of lpod2asp(Π)base, and let

shrink(S, x1, . . . , xm) be {a(v) | a(v, x1, . . . , xm) ∈ S and a(v) ∈ σ}.

If S |= ap(x1, . . . , xm), we define the set shrink(S, x1, . . . , xm) as a candidate answer set on
σ of lpod2asp(Π)base.3

The following proposition asserts the soundness of the translation lpod2asp(Π)base.

Proposition 2
The candidate answer sets of an LPOD Π of signature σ are exactly the candidate answer sets on
σ of lpod2asp(Π)base.

Example 1 Continued: The following is the encoding of lpod2asp(Π1)base in the input language
of CLINGO.

%%%% 1 %%%%
{ap(X1,X2): X1=0..2, X2=0..2}. :∼ ap(X1,X2). [-1, X1, X2]

%%%% 2 %%%%
% a*b <- not c.
body_1(X1,X2) :- ap(X1,X2), not c(X1,X2).
:- ap(X1,X2), X1=0, body_1(X1,X2). :- ap(X1,X2), X1>0, not body_1(X1,X2).

a(X1,X2) :- body_1(X1,X2), X1=1. b(X1,X2) :- body_1(X1,X2), X1=2.

:- body_1(X1,X2), X1!=1, a(X1,X2).
:- body_1(X1,X2), X1!=2, not a(X1,X2), b(X1,X2).

% b*c <- not d.
body_2(X1,X2) :- ap(X1,X2), not d(X1,X2).
:- ap(X1,X2), X2=0, body_2(X1,X2). :- ap(X1,X2), X2>0, not body_2(X1,X2).

b(X1,X2) :- body_2(X1,X2), X2=1. c(X1,X2) :- body_2(X1,X2), X2=2.

:- body_2(X1,X2), X2!=1, b(X1,X2).
:- body_2(X1,X2), X2!=2, not b(X1,X2), c(X1,X2).

%%%% 3 %%%%
1{degree(ap(X1,X2), D1, D2): D1=1..2, D2=1..2}1 :- ap(X1,X2).

:- degree(ap(X1,X2), D1, D2), X1=0, D1!=1.
:- degree(ap(X1,X2), D1, D2), X1>0, D1!=X1.

:- degree(ap(X1,X2), D1, D2), X2=0, D2!=1.
:- degree(ap(X1,X2), D1, D2), X2>0, D2!=X2.

3 We also apply this notation to the full translation lpod2asp(Π) and crp2asp(Π) below.

The optimal answer set S of lpod2asp(Π1)base is

{ap(1, 1), a(1, 1), b(1, 1), . . . , ap(2, 1), b(2, 1), . . . , ap(0, 2), c(0, 2), . . . } (21)

(bodyi(·) and degree(·) atoms are not listed). Since S satisfies ap(1, 1), ap(2, 1), and ap(0, 2),
the candidate answer sets on σ of lpod2asp(Π1)base are

shrink(S, 1, 1) = {a, b}, shrink(S, 2, 1) = {b}, shrink(S, 0, 2) = {c}

which are exactly the candidate answer sets of Π1.

2.3.2 Find Preferred Answer Sets

The second part of the translation lpod2asp(Π) is to compare the candidate answer sets to find the
preferred answer sets. For each preference criterion, lpod2asp(Π) contains the following rules
respectively. Below maxdegree is max{ni | i ∈ {1, . . . ,m}}.

(a) Cardinality-Preferred: For this criterion, lpod2asp(Π) contains the following rules.

card(P,X,N) ← degree(P,D1, . . . , Dm), X = 1..maxdegree,

N = {D1 = X; . . . ;Dm = X}. (22)

equ2degree(P1, P2, X) ← card(P1, X,N), card(P2, X,N), P1 6= P2. (23)

prf2degree(P1, P2, X) ← card(P1, X,N1), card(P2, X,N2), N1 > N2. (24)

prf (P1, P2) ← X = 0..maxdegree− 1, prf2degree(P1, P2, X + 1),

X{equ2degree(P1, P2, Y ) : Y = 1..X}. (25)

pAS(X1, . . . , Xm) ← ap(X1, . . . , Xm), {prf (P, ap(X1, . . . , Xm))}0. (26)

P ,P1, andP2 denote assumption programs in the form of ap(X1, . . . , Xm). card(P,X,N)

is true if P satisfies N rules in Πod to degree X . equ2degree(P1, P2, X) is true if P1 and
P2 have the same number of rules that are satisfied to degree X . prf2degree(P1, P2, X)

is true if P1 satisfies more rules to degree X than P2 does. prf (P1, P2) is true if P1 is
cardinality-preferred to P2: P1 satisfies more rules to degree X + 1 than P2 does whereas
they satisfy the same number of rules up to degreeX . Rule (26) reads as: given an assump-
tion program represented by ap(X1, . . . , Xm), if we cannot find an assumption program P

that is more preferable, then the answer sets of ap(X1, . . . , Xm) are all preferred answer
sets of Π. Note that P in rule (26) is a local variable that ranges over all ap(·) atoms.

(b) Inclusion-Preferred: For this criterion, lpod2asp(Π) contains the following rules.

even(0; 2). (27)

equ2degree(P1, P2, X) ← P1 6= P2, X = 1..maxdegree,

degree(P1, D11, . . . , D1m), degree(P2, D21, . . . , D2m),

C1 = {D11 = X;D21 = X}, . . . , Cm = {D1m = X;D2m = X},
even(C1), . . . , even(Cm). (28)

prf2degree(P1, P2, X) ← P1 6= P2, X = 1..maxdegree,

not equ2degree(P1, P2, X),

degree(P1, D11, . . . , D1m), degree(P2, D21, . . . , D2m),

{D11 6= X;D21 = X}1, . . . , {D1m 6= X;D2m = X}1. (29)

prf (P1, P2) ← X = 0..maxdegree− 1, prf2degree(P1, P2, X + 1),

X{equ2degree(P1, P2, Y ) : Y = 1..X}. (30)

pAS(X1, . . . , Xm) ← ap(X1, . . . , Xm), {prf (P, ap(X1, . . . , Xm))}0. (31)

where {D11 = X;D21 = X} counts the number of true atoms in this set, so it equals to
0 (or 2) when none (or both) of D11 = X and D21 = X are true; {D11 6= X;D21 = X}1
means that the number of true atoms in this set must be smaller or equal to 1, which means
that D11 6= X and D21 = X cannot be true at the same time – in other words, D21 = X

implies D11 = X .
(c) Pareto-Preferred: For this criterion, lpod2asp(Π) contains the following rules.

equ(P1, P2) ← degree(P1, D1, . . . , Dm), degree(P2, D1, . . . , Dm). (32)

prf (P1, P2) ← degree(P1, D11, . . . , D1m), degree(P2, D21, . . . , D2m),

not equ(P1, P2), D11 ≤ D21, . . . , D1m ≤ D2m. (33)

pAS(X1, . . . , Xm) ← ap(X1, . . . , Xm), {prf (P, ap(X1, . . . , Xm))}0. (34)

where equ(P1, P2) means that P1 is equivalent to P2 at all degrees.
(d) Penalty-Sum-Preferred: For this criterion, lpod2asp(Π) contains the following rules.

sum(P,N) ← degree(P,D1, . . . , Dm), N = D1 + · · ·+Dm. (35)

prf (P1, P2) ← sum(P1, N1), sum(P2, N2), N1 < N2. (36)

pAS(X1, . . . , Xm) ← ap(X1, . . . , Xm), {prf (P, ap(X1, . . . , Xm))}0. (37)

where sum(P,N) means that the sum of P ’s satisfaction degrees of all rules is N .

If S |= pAS(x1, . . . , xm), we define the set shrink(S, x1, . . . , xm) to be a preferred answer
set on σ of lpod2asp(Π).

a
The following theorem assert the soundness of the translation lpod2asp(Π).

Theorem 1
Under any of the four preference criteria, the candidate (preferred, respectively) answer sets of
an LPOD Π of signature σ are exactly the candidate (preferred, respectively) answer sets on σ of
lpod2asp(Π).

Example 2 Continued: The first part of lpod2asp(Π2) contains the following rules.

#const maxdegree = 4.

%%%% 1 %%%%

{ap(X1,X2): X1=0..4, X2=0..3}. :∼ ap(X1,X2). [-1, X1, X2]

%%%% 2 %%%%

1{hotel(H,X1,X2): H=1..3}1 :- ap(X1,X2).
:- ap(X1,X2), hotel(1,X1,X2), not close(X1,X2).
:- ap(X1,X2), hotel(1,X1,X2), not star2(X1,X2).
:- ap(X1,X2), hotel(2,X1,X2), not med(X1,X2).
:- ap(X1,X2), hotel(2,X1,X2), not star3(X1,X2).
:- ap(X1,X2), hotel(3,X1,X2), not tooFar(X1,X2).
:- ap(X1,X2), hotel(3,X1,X2), not star4(X1,X2).

% close * med * far * tooFar.

body_1(X1,X2) :- ap(X1,X2).

:- ap(X1,X2), X1=0, body_1(X1,X2). :- ap(X1,X2), X1>0, not body_1(X1,X2).

close(X1,X2) :- body_1(X1,X2), X1=1. med(X1,X2) :- body_1(X1,X2), X1=2.
far(X1,X2) :- body_1(X1,X2), X1=3. tooFar(X1,X2) :- body_1(X1,X2), X1=4.

:- body_1(X1,X2), X1!=1, close(X1,X2).
:- body_1(X1,X2), X1!=2, not close(X1,X2), med(X1,X2).
:- body_1(X1,X2), X1!=3, not close(X1,X2), not med(X1,X2), far(X1,X2).
:- body_1(X1,X2), X1!=4, not close(X1,X2), not med(X1,X2), not far(X1,X2),

tooFar(X1,X2).

% star4 * star3 * star2.

body_2(X1,X2) :- ap(X1,X2).

:- ap(X1,X2), X2=0, body_2(X1,X2). :- ap(X1,X2), X2>0, not body_2(X1,X2).

star4(X1,X2) :- body_2(X1,X2), X2=1. star3(X1,X2) :- body_2(X1,X2), X2=2.
star2(X1,X2) :- body_2(X1,X2), X2=3.

:- body_2(X1,X2), X2!=1, star4(X1,X2).
:- body_2(X1,X2), X2!=2, not star4(X1,X2), star3(X1,X2).
:- body_2(X1,X2), X2!=3, not star4(X1,X2), not star3(X1,X2), star2(X1,X2).

%%%% 3 %%%%

1{degree(ap(X1,X2), D1, D2): D1=1..4, D2=1..3}1 :- ap(X1,X2).

:- degree(ap(X1,X2), D1, D2), X1=0, D1!=1.
:- degree(ap(X1,X2), D1, D2), X1>0, D1!=X1.

:- degree(ap(X1,X2), D1, D2), X2=0, D2!=1.
:- degree(ap(X1,X2), D1, D2), X2>0, D2!=X2.

For the second part of the translation, lpod2asp(Π2) contains one of the following sets of rules.

%%%% a. Cardinality %%%%
card(P,X,N) :- degree(P,D1,D2), X=1..maxdegree, N={D1=X; D2=X}.
equ2degree(P1,P2,X) :- card(P1,X,N), card(P2,X,N), P1!=P2.
prf2degree(P1,P2,X) :- card(P1,X,N1), card(P2,X,N2), N1>N2.
prf(P1,P2) :- X=0..maxdegree-1, prf2degree(P1,P2,X+1), X{equ2degree(P1,P2,Y): Y=1..X}.
pAS(X1,X2) :- ap(X1,X2), {prf(P, ap(X1,X2))}0.

%%%% b. Inclusion %%%%
even(0;2).
equ2degree(P1,P2,X) :- P1!=P2, X=1..maxdegree, degree(P1,D11,D12), degree(P2,D21,D22),

C1 = {D11=X; D21=X}, C2={D12=X; D22=X}, even(C1), even(C2).
prf2degree(P1,P2,X) :- P1!=P2, X=1..maxdegree, not equ2degree(P1,P2,X),

degree(P1,D11,D12), degree(P2,D21,D22),
{D11!=X; D21=X}1, {D12!=X; D22=X}1.

prf(P1,P2) :- X=0..maxdegree-1, prf2degree(P1,P2,X+1), X{equ2degree(P1,P2,Y): Y=1..X}.
pAS(X1,X2) :- ap(X1,X2), {prf(P, ap(X1,X2))}0.

%%%% c. Pareto %%%%
equ(P1,P2) :- degree(P1,D1,D2), degree(P2,D1,D2).
prf(P1,P2) :- degree(P1,D11,D12), degree(P2,D21,D22), not equ(P1,P2),

D11<=D21, D12<=D22.
pAS(X1,X2) :- ap(X1,X2), {prf(P, ap(X1,X2))}0.

%%%% d. Penalty-Sum %%%%
sum(P,N) :- degree(P,D1,D2), N=D1+D2.
prf(P1,P2) :- sum(P1,N1), sum(P2,N2), N1<N2.
pAS(X1,X2) :- ap(X1,X2), {prf(P, ap(X1,X2))}0.

Note that each set of rules in the second part conservatively extends the answer set of the base
program. For example, the optimal answer set of lpod2asp(Π1) under Penalty-Sum preference is
the union of (21) and {sum(ap(0, 2), 3), sum(ap(1, 1), 2), sum(ap(2, 1), 3), prf(ap(1, 1), ap(0, 2)),
prf(ap(1, 1), ap(2, 1)), pAS(1, 1)}, which indicates that {a, b} is the preferred answer set.

The optimal answer set S of lpod2asp(Π2) under the cardinality preference is

{pAS(1, 3), ap(1, 3), hotel(1, 1, 3), close(1, 3), star2(1, 3),

ap(2, 2), hotel(2, 2, 2), med(2, 2), star3(2, 2),

ap(4, 1), hotel(3, 4, 1), tooFar(4, 1), star4(4, 1), . . . }

Since S satisfies ap(1, 3), ap(2, 2), and ap(4, 1), the candidate answer sets on σ of lpod2asp(Π2)

are

shrink(S, 1, 3) = {hotel(1), close, star2},
shrink(S, 2, 2) = {hotel(2),med, star3},
shrink(S, 4, 1) = {hotel(3), tooFar, star4},

which are exactly the candidate answer sets of Π2. Since S satisfies pAS(1, 3), the preferred an-
swer sets on σ of lpod2asp(Π2) is shrink(S, 1, 3) = {hotel(1), close, star2} which is exactly
the cardinality-preferred answer set of Π2. Let

pAS1 = {pAS(1, 3), hotel(1, 1, 3), close(1, 3), star2(1, 3)},
pAS2 = {pAS(2, 2), hotel(2, 2, 2),med(2, 2), star3(2, 2)},
pAS3 = {pAS(4, 1), hotel(3, 4, 1), tooFar(4, 1), star4(4, 1)}.

The optimal answer sets of lpod2asp(Π2) under 4 criteria contain

cardinality-preferred: pAS1 inclusion-preferred: pAS1 ∪ pAS3

Pareto-preferred: pAS1 ∪ pAS2 ∪ pAS3 penalty-sum-preferred: pAS1 ∪ pAS2

which are in a 1-1 correspondence with the preferred answer sets of Π2 under each of the four
criteria respectively.

3 CR-Prolog2 to ASP with Weak Constraints

3.1 Review: CR-Prolog2

We review the definition of CR-Prolog2 from (Balduccini et al. 2003).

Syntax: A (propositional) CR-Prolog2 program Π consists of four kinds of rules:

regular rule Head ← Body (38)

ordered rule i : C1 × · · · × Cni ← Body (39)

cr-rule i : Head +← Body (40)

ordered cr-rule i : C1 × · · · × Cni
+← Body (41)

where Head ← Body is a standard ASP rule, i is the index of the rule, Cj are atoms, and ni ≥ 2.
The intuitive meaning of an ordered disjunctionC1×· · ·×Cni is similar to the one for LPOD. A
cr-rule (40) or an ordered cr-rule (41) is applied in Π if it is treated as a usual ASP rule in Π (by
replacing +← with←); it is not applied if it is omitted in Π. A cr-rule (40) or an ordered cr-rule
(41) is applied only if the agent has no way to obtain a consistent set of beliefs using regular
rules or ordered rules only. By Head(i) and Body(i), we denote the head and the body of rule i.

Semantics: The semantics of CR-Prolog2 is based on the transformation from a CR-Prolog2

program Π of signature σ into an answer set program HΠ, which is constructed as follows. The
first-order signature of HΠ is σ ∪ {choice/2, appl/1, fired/1, isPreferred/2}, where choice is a
function constant, appl, fired, isPreferred are predicate constants not in σ.

1. Let RΠ be the set of rules obtained from Π by replacing every cr-rule and ordered cr-rule
of index i with a rule:

i : Head(i)← Body(i), appl(i)

where appl(i) means rule i is applied. Notice that RΠ contains only regular rules and
ordered rules.
HΠ is then obtained fromRΠ by replacing every ordered rule of index r, where Head(r) =

C1 × · · · × Cni , with the following rules (for 1 ≤ j ≤ ni):

Cj ← Body(r), appl(choice(r, j))

fired(r)← appl(choice(r, j))

prefer(choice(r, j), choice(r, j + 1)) (j < ni)

⊥ ← Body(r), not fired(r)

(42)

where appl(choice(r, j)) means that the j-th atom in the ordered disjunction Head(r) is
chosen, i.e., Cj is true if Head(r) is true.

2. HΠ also contains the following set of rules:

isPreferred(R1, R2)← prefer(R1, R2).

isPreferred(R1, R3)← prefer(R1, R2), isPreferred(R2, R3).

⊥ ← isPreferred(R,R).

⊥ ← appl(R1), appl(R2), isPreferred(R1, R2).

where R1, R2, R3 are schematic variables ranging over indices of cr-rules and ordered
cr-rules in Π as well as terms of the form choice(·).

By atoms(HΠ, {appl}), we denote the set of atoms in HΠ in the form of appl(·). A general-
ized answer set of Π is an answer set of HΠ ∪A where A ⊆ atoms(HΠ, {appl}).

Let S1, S2 be generalized answer sets of Π. S1 dominates S2 if there exist r1 and r2 such
that appl(r1) ∈ S1, appl(r2) ∈ S2, and isPreferred(r1, r2) ∈ S1 ∩ S2. Further, we say this
domination is rule-wise if r1 and r2 are indices of two cr-rules; atom-wise if r1 and r2 are two

terms of the form choice(·). S1 is a candidate answer set of Π if there is no other generalized
answer set that dominates S1.

The projection of S1 onto σ is a preferred answer set of Π if S1 is a candidate answer set of Π

and there is no other candidate answer set S2 such that S2 ∩ atoms(HΠ, {appl}) ⊂ S1.

Example 3
(From (Balduccini et al. 2003)) Consider the following CR-Prolog2 program Π3:

q ← t.

s← t.

p← not q.
r ← not s.
← p, r.

1 : t
+← .

2 : q × s +← .

which has 5 generalized answer sets (the atoms formed by isPreferred or fired are omitted)

S1 = {q, s, t, appl(1), prefer(choice(2, 1), choice(2, 2))}
S2 = {q, r, appl(2), appl(choice(2, 1)), prefer(choice(2, 1), choice(2, 2))}
S3 = {p, s, appl(2), appl(choice(2, 2)), prefer(choice(2, 1), choice(2, 2))}
S4 = {q, s, t, appl(1), appl(2), appl(choice(2, 1)), prefer(choice(2, 1), choice(2, 2))}
S5 = {q, s, t, appl(1), appl(2), appl(choice(2, 2)), prefer(choice(2, 1), choice(2, 2))}.

Since S2 (atom-wise) dominates S3 and S5, the candidate answer sets are S1, S2, and S4. Since
S1 ∩ atoms(HΠ3

, {appl}) ⊂ S4, the preferred answer sets of Π3 are the projections from S1 or
S2 onto σ.

3.2 Turning CR-Prolog2 into ASP with Weak Constraints

We define a translation crp2asp(Π) that turns a CR-Prolog2 program Π into an answer set pro-
gram with weak constraints.

Let Π be a CR-Prolog2 program of signature σ, where its rules are rearranged such that the
cr-rules are of indices 1, . . . , k, the ordered cr-rules are of indices k + 1, . . . , l, and the ordered
rules are of indices l + 1, . . . ,m.

For an ordered rule (39) or an ordered cr-rule (41), its i-th assumption, where i ∈ {1, . . . , ni},
is defined as Ci ← Body. An assumption program AP (x1, . . . , xm) of Π whose assumption
degree list is (x1, . . . , xm) is obtained from Π as follows (xi ∈ {0, 1} if i = 1, . . . k; xi ∈
{0, . . . , ni} if i = k+1, . . . , l; xi ∈ {1, . . . , ni} if i = l+1, . . . ,m, where ni is the number of
atoms in the head of rule i).

• every regular rule (38) is in AP (x1, . . . , xm);
• a cr-rule (40) is omitted if xi = 0, and is replaced by Head ← Body if xi = 1;
• an ordered cr-rule (41) is omitted if xi = 0, and is replaced by its xi-th assumption if
xi > 0;

• an ordered rule (39) is replaced by its xi-th assumption.

Besides, each assumption program AP (x1, . . . , xm) contains

isPreferred(R1, R2)← prefer(R1, R2).

isPreferred(R1, R3)← prefer(R1, R2), isPreferred(R2, R3).

← isPreferred(R,R).

← xr1 > 0, xr2 > 0, isPreferred(r1, r2). (1 ≤ r1, r2 ≤ l)

The generalized answer sets of Π can be obtained from the answer sets of all the assumption
programs of Π.

Proposition 3
For any CR-Prolog2 program Π of signature σ, a setX of atoms is the projection of a generalized
answer set of Π onto σ iff X is the projection of an answer set of an assumption program of Π

onto σ.

Let Π1 and Π2 be two assumption programs of Π. We say an answer set S1 of Π1 dominates
an answer set S2 of Π2 if (i) there exists a rule i in Π that is replaced by its j1-th assumption
in Π1, is replaced by its j2-th assumption in Π2, and j1 < j2; or (ii) there exist 2 rules r1, r2

in Π such that r1 is applied in Π1, r2 is applied in Π2, and prefer(r1, r2) ∈ S1 ∩ S2. Indeed,
by Proposition 3, S1 dominates S2 iff the corresponding generalized answer set of the former
dominates that of the latter.

An answer set program with weak constraints crp2asp(Π) is obtained from Π based on the
notion of assumption programs as follows. The first-order signature σ′ of crp2asp(Π) contains
m-ary predicate constant a/m for each propositional constant a of σ. Besides, σ′ contains the
following predicate constants not in σ: ap/m, dominate/2, isPreferred/(m+2), candidate/m,
lessCrRulesApplied/2, and pAS/m.

1. To consider a maximal set of consistent assumption programs, crp2asp(Π) contains

{ap(X1, . . . , Xm) : X1 = 0..1, . . . , Xk = 0..1, Xk+1 = 0..nk+1, . . . , Xl = 0..nl,

Xl+1 = 1..nl+1, . . . , Xm = 1..nm}. (43)

:∼ ap(X1, . . . , Xm). [−1, X1, . . . , Xm] (44)

where ni is the number of atoms in Head(i), ap(X1, . . . , Xp) denotes an assumption program
obtained from Π.

2. crp2asp(Π) contains the following rules to construct all assumption programsAP (x1, . . . , xm):

• for each regular rule Head ← Body in Π, crp2asp(Π) contains

Head(X1, . . . , Xm)← ap(X1, . . . , Xm), Body(X1, . . . , Xm) (45)

• for each cr-rule i : Headi
+← Bodyi in Π, crp2asp(Π) contains

Headi(X1, . . . , Xm)← ap(X1, . . . , Xm),Bodyi(X1, . . . , Xm), Xi = 1 (46)

• for each ordered rule or ordered cr-rule i : C1
i × · · · × Cn

i

(+)← Bodyi in Π, for
1 ≤ j ≤ ni, crp2asp(Π) contains

Cj
i (X1, . . . , Xm)← ap(X1, . . . , Xm),Bodyi(X1, . . . , Xm), Xi = j (47)

3. To define dominate in the semantics of CR-Prolog2, crp2asp(Π) contains the following rules.

Atom-wise dominance: Instead of using choice(·) terms and appl(choice(·)) atoms in (42), we
represent the atom wise dominance by comparing the assumption degrees. For ordered cr-rules
and ordered rules i ∈ {k + 1, . . .m}, we include

dominate(ap(X1, . . . , Xm), ap(Y1, . . . , Ym))←
ap(X1, . . . , Xm), ap(Y1, . . . , Ym), 0 < Xi, Xi < Yi (48)

rule-wise dominance: The following rules are included only when Π contains an atom prefer(·).

r1 and r2 ranges over {1, . . . , l}.
isPreferred(R1, R2, X1, . . . , Xm)← prefer(R1, R2, X1, . . . , Xm) (49)

isPreferred(R1, R3, X1, . . . , Xm)← prefer(R1, R2, X1, . . . , Xm),

isPreferred(R2, R3, X1, . . . , Xm) (50)

← isPreferred(R,R,X1, . . . , Xm) (51)

← isPreferred(r1, r2, X1, . . . , Xm), Xr1 > 0, Xr2 > 0 (52)

dominate(ap(X1, . . . , Xm), ap(Y1, . . . , Ym))← ap(X1, . . . , Xm), ap(Y1, . . . , Ym),

isPreferred(r1, r2, X1, . . . , Xm), isPreferred(r1, r2, Y1, . . . , Ym), Xr1 > 0, Yr2 > 0 (53)

We say an assumption program Π1 dominates an assumption program Π2 if an answer set
of Π1 dominates an answer set of Π2. Indeed, our translation guarantees that if Π1 dominates
Π2, all answer sets of Π1 dominates any answer sets of Π2. Rule (48) says that the assumption
program AP (x1, . . . , xm) dominates the assumption program AP (y1, . . . , ym) if there exists a
rule i in Π that is replaced by its xi-th assumption in AP (x1, . . . , xm), by its yi-th assumption
in AP (y1, . . . , ym), and xi < yi. Rules (49), (50), (51), (52) are the set of rules in the semantics
of CR-Prolog2 with the extended signature σ′. Rule (53) says that AP (x1, . . . , xm) dominates
AP (y1, . . . , ym) if isPreferred(r1, r2) is true in both assumption programs while r1 is applied
in AP (x1, . . . , xm) and r2 is applied in AP (y1, . . . , ym).

4. To define candidate answer sets in the semantics of CR-Prolog2, crp2asp(Π) contains

candidate(X1, . . . , Xm)← ap(X1, . . . , Xm), {dominate(P, ap(X1, . . . , Xm))}0 (54)

Rule (54) says that the answer sets of AP (x1, . . . , xm) are candidate answer sets if there does
not exist an assumption program P that dominates AP (x1, . . . , xm).

5. To define the preference between two candidate answer sets and find preferred answer sets,
crp2asp(Π) contains

lessCrRulesApplied(ap(X1, . . . , Xm), ap(Y1, . . . , Ym))←
candidate(X1, . . . , Xm), candidate(Y1, . . . , Ym),

1{X1 6= Y1; . . . ;Xm 6= Ym}, X1 ≤ Y1, . . . , Xm ≤ Ym (55)

pAS(X1, . . . , Xm)← candidate(X1, . . . , Xm), {lessCrRulesApplied(P, ap(X1, . . . , Xm))}0
(56)

Rule (55) says that for any different assumption programsAP (x1, . . . , xm) andAP (y1, . . . , ym)

whose answer sets are candidate answer sets, if all the choices in AP (x1, . . . , xm) is not worse
than 4 those inAP (y1, . . . , ym), then the former must apply less cr-rules or ordered cr-rules than
the latter. Rule (56) says that the answer sets of AP (x1, . . . , xm) are preferred answer sets if
these answer sets are candidate answer sets and there does not exist an assumption program P

that applies less cr-rules than AP (x1, . . . , xm).
Let S be an optimal answer set of crp2asp(Π); x1, . . . , xm be a list of integers. If S |=

ap(x1, . . . , xm), we define the set shrink(S, x1, . . . , xm) as a generalized answer set on σ of
crp2asp(Π); if S |= candidate(x1, . . . , xm), we define the set shrink(S, x1, . . . , xm) as a can-
didate answer set on σ of crp2asp(Π); if S |= pAS(x1, . . . , xp), we define the set shrink(S, x1, . . . , xp)

as a preferred answer set on σ of crp2asp(Π).

4 i.e., for any rule i in Π, if it is applied in AP (x1, . . . , xm), it must be applied in AP (y1, . . . , ym); if it is replaced
by its xi-th assumption in AP (x1, . . . , xm), it must be replaced by its yi-th assumption in AP (y1, . . . , ym) and
xi ≤ yi

Theorem 2
For any CR-Prolog2 program Π of signature σ, (a) the projections of the generalized answer sets
of Π onto σ are exactly the generalized answer sets on σ of crp2asp(Π). (b) the projections of
the candidate answer sets of Π onto σ are exactly the candidate answer sets on σ of crp2asp(Π).
(c) the preferred answer sets of Π are exactly the preferred answer sets on σ of crp2asp(Π).

Example 3 Continued: The translated ASP program crp2asp(Π3) is

%%%% 1 %%%%
{ap(X1,X2): X1=0..1, X2=0..2}. :∼ ap(X1,X2). [-1,X1,X2]

%%%% 2 %%%%
q(X1,X2) :- ap(X1,X2), t(X1,X2). s(X1,X2) :- ap(X1,X2), t(X1,X2).
p(X1,X2) :- ap(X1,X2), not q(X1,X2). r(X1,X2) :- ap(X1,X2), not s(X1,X2).
:- ap(X1,X2), p(X1,X2), r(X1,X2).

% 1: t <+-.
t(X1,X2) :- ap(X1,X2), X1=1.

% 2: q*s <+-.
q(X1,X2) :- ap(X1,X2), X2=1. s(X1,X2) :- ap(X1,X2), X2=2.

%%%% 3 %%%%
dominate(ap(X1,X2), ap(Y1,Y2)) :- ap(X1,X2), ap(Y1,Y2), 0<X1, X1<Y1.
dominate(ap(X1,X2), ap(Y1,Y2)) :- ap(X1,X2), ap(Y1,Y2), 0<X2, X2<Y2.

%%%% 4 %%%%
candidate(X1,X2) :- ap(X1,X2), {dominate(P,ap(X1,X2))}0.

%%%% 5 %%%%
lessCrRulesApplied(ap(X1,X2), ap(Y1,Y2)) :- candidate(X1,X2), candidate(Y1,Y2),

1{X1!=Y1;X2!=Y2}, X1<=Y1, X2<=Y2.
pAS(X1,X2) :- candidate(X1,X2), {lessCrRulesApplied(P,ap(X1,X2))}0.

The optimal answer set S of crp2asp(Π3) is

{pAS(1, 0), candidate(1, 0), ap(1, 0), t(1, 0), q(1, 0), s(1, 0),

pAS(0, 1), candidate(0, 1), ap(0, 1), q(0, 1), r(0, 1),

ap(0, 2), p(0, 2), s(0, 2),

candidate(1, 1), ap(1, 1), t(1, 1), q(1, 1), s(1, 1),

ap(1, 2), t(1, 2), q(1, 2), s(1, 2), . . . }.

Since S satisfies ap(1, 0), ap(0, 1), ap(0, 2), ap(1, 1), ap(1, 2), the generalized answer sets on σ
of crp2asp(Π3) are

shrink(S, 1, 0) = {t, q, s}
shrink(S, 0, 1) = {q, r}
shrink(S, 0, 2) = {p, s}

shrink(S, 1, 1) = {t, q, s}
shrink(S, 1, 2) = {t, q, s}

which are exactly the projections of the generalized answer sets of Π3 onto σ. Similarly, we
observe that the candidate (preferred, respectively) answer sets on σ of crp2asp(Π3) are exactly
the projections of the candidate (preferred, respectively) answer sets of Π3 onto σ.

Furthermore, let Π′3 = Π3∪{prefer(2, 1).}. The translation crp2asp(Π′3) is crp2asp(Π3)∪R,
where R is the set of the following rules:

%%%% 2 %%%%
prefer(2,1,X1,X2) :- ap(X1,X2).

%%%% 3 %%%%
isPreferred(R1,R2,X1,X2) :- prefer(R1,R2,X1,X2).
isPreferred(R1,R3,X1,X2) :- prefer(R1,R2,X1,X2), isPreferred(R2,R3,X1,X2).
:- isPreferred(R,R,X1,X2).
:- isPreferred(2,1,X1,X2), X2>0, X1>0.

dominate(ap(X1,X2), ap(Y1,Y2)) :- ap(X1,X2), ap(Y1,Y2),
isPreferred(2,1,X1,X2), isPreferred(2,1,Y1,Y2), X2>0, Y1>0.

The optimal answer set S of crp2asp(Π′3) is

{ ap(1, 0), t(1, 0), q(1, 0), s(1, 0),

pAS(0, 1), candidate(0, 1), ap(0, 1), q(0, 1), r(0, 1),

ap(0, 2), p(0, 2), s(0, 2), . . . }

and it is easy to check that the generalized (/candidate/preferred) answer sets on σ of crp2asp(Π′3)

are exactly the projections of the generalized (/candidate/preferred) answer sets of Π′3 onto σ.

4 Related Work and Conclusion

We presented reductions of LPOD and CR-Prolog2 into the standard ASP language, which ex-
plains the new constructs for preference handling in terms of the standard ASP language. The
one-pass translations are theoretically interesting. They may be a useful tool for studying the
mathematical properties of LPOD and CR-Prolog2 programs by reducing them to more well-
known properties of standard answer set programs. Both translations are “almost” modular in
the sense that the translations are rule-by-rule but the argument of each atom representing the
assumption degrees may need to be expanded when new rules are added.

However, the direct implementations may not lead to effective implementations. The size of
lpod2asp(Π) and crp2asp(Π) after grounding could be exponential to the size of the non-regular
rules in Π. This is because these translations compare all possible assumption programs whose
number is exponential to the size of non-regular rules. One may consider parallelizing the com-
putation of assumption programs since they are disjoint from each other according to the trans-
lations.

In a sense, our translations are similar to the meta-programming approach to handle preference
in ASP (e.g., (Delgrande et al. 2003)) in that we turn LPOD and CR-Prolog2 into answer set
programs that do not have the built-in notion of preference.

In (Brewka et al. 2002), LPOD is implemented using SMODELS. The implementation inter-
leaves the execution of two programs–a generator which produces candidate answer sets and a
tester which checks whether a given candidate answer set is maximally preferred or produces a
more preferred candidate if it is not. An implementation of CR-Prolog reported in (Balduccini
2007) uses a similar algorithm. In contrast, the reductions shown in this paper can be computed
by calling an answer set solver one time without the need for iterating the generator and the tester.
This feature may be useful for debugging LPOD and CR-Prolog2 programs because it allows us
to compare all candidate and preferred answer sets globally.

Asprin (Brewka et al. 2015) provides a flexible way to express various preference relations

over answer sets and is implemented in CLINGO. Similar to the existing LPOD solvers, CLINGO

makes iterative calls to find preferred answer sets, unlike the one-shot execution as we do.
Asuncion et al. (2014) presents a first-order semantics of logic programs with ordered disjunc-

tion by translation into second-order logic whereas our translation is into the standard answer set
programs.

